# Copyright 1999-2010 Gentoo Foundation
# Distributed under the terms of the GNU General Public License v2
# $Header: $

EAPI=2

inherit toolchain-funcs

MY_PV=${PV/_beta/b}
MY_P=${PN}-${MY_PV}

DESCRIPTION="Web application security scanner"
HOMEPAGE="http://code.google.com/p/skipfish/"
SRC_URI="http://skipfish.googlecode.com/files/${MY_P}.tgz"

LICENSE="Apache-2.0"
SLOT="0"
KEYWORDS="~x86 ~amd64"
IUSE="debug"

DEPEND="dev-libs/openssl
	net-dns/libidn
	sys-libs/zlib"
RDEPEND="${DEPEND}"

S=${WORKDIR}/${PN}

src_prepare() {
	# Change assets and default directory paths
	sed -i \
		-e "/ASSETS_DIR/s:assets:/usr/share/doc/${PF}/html:" \
		-e "/DEF_WORDLIST/s:skipfish.wl:/usr/share/${PN}/dictionaries/default.wl:" \
		config.h || die "sed failed"
}

src_compile() {
	tc-export CC

	local _debug
	use debug && _debug=debug

	emake ${_debug} || die "emake failed"
}

src_install() {
	dobin ${PN}

	insinto /usr/share/${PN}/dictionaries
	doins dictionaries/*.wl

	dohtml assets/*

	dodoc README dictionaries/README-FIRST ChangeLog
}

pkg_postinst() {
	elog
	elog "Check out /usr/share/${PF}/dictionaries/ for other dictionaries"
	elog "See README-FIRST in /usr/share/doc/${PF}/ to pick a dictionary" 
	elog
	elog "Having problems with your scans? Be sure to visit:"
	elog " http://code.google.com/p/skipfish/wiki/KnownIssues"
	elog
}
